Program speeds planning process for crude pipeline

Thomas A. Dickman
Pure Transportation Co.
Schaumburg, Ill.

The hydraulic examination of existing proposed crude oil pipelines involves many repetitive calculations as well as hydraulic profile drawings. To expedite this process, Union Oil Co.'s Schaumburg pipeline department has created a computer program to perform the calculations and prepare the drawings.

This article discusses that program and explains its operation.

Calculations. In project planning, a portion of the task is to find the optimum line diameter, route, number and location of pump stations, and the power requirements at each station.

As with any discussion of fluid flow, it is clear one must begin with the physical properties of the fluid.

It is from these parameters that the behavior of the system can be projected for a variety of situations.

In the gross assay, it must be determined if the crude oil exhibits an efflux time in seconds of one hundred or less as determined through a short tube viscometer for viscous oils and expressed in Saybolt Universal Seconds (SUS).

This temperature-dependent value as it stands is useful only to collate among other fluids. Its application in predicting behavior exists only after employment in determining the fluid's kinematic viscosity (KV), that is, the distribution of the absolute viscosity (μ) over the average density (ρ) of the crude oil:

$$KV = \frac{\mu}{\rho}$$

A vast amount of research has been conducted on the relationship of these properties. Two empirical formulas have been determined which incorporate average viscosimeter constants and are accurate to within 2% over a wide viscosity range:

$$KV = 0.0226x - 1.95/x$$

Where: x = Efflux time, 32-100 SUS

$$KV = 0.00220x - 1.35/x$$

Where: x = Efflux time, more than 100 SUS

The energy lost to friction due to the movement of fluid through a pipe is influenced by the condition of flow. For laminar flow, the roughness or condition of the pipe's interior surface has little effect.

Empirical studies indicate that when the dimensionless Reynolds number is below the critical value of 2,200, flow can be taken as laminar. The friction factor is inversely proportional to the Reynolds number as indicated in the following:

$$f = \frac{64}{Re}$$

Where:

f = Friction factor, dimensionless
Re = Reynolds number, dimensionless

Recognizing that friction loss depends on the length of pipe, internal diameter, velocity of the liquid and its viscosity, in addition to the friction factor, the semi-theoretical Fanning formula can determine the head loss for a given length of level pipe.

$$H = f\frac{LV2}{2k\ D}$$

Where:

H = Friction head loss, ft
f = Friction factor, dimensionless
L = Length of pipe, ft
D = Inside diameter (ID) of pipe, ft
V = Fluid velocity, fps
k = Acceleration due to gravity, fps

Flow regime. Numerous investigators have studied the flow of fluids in pipes confirming the dependency of the flow regime on the Reynolds number. Between the values of 2,200 and 4,000, the flow passes through a transitional stage from laminar to turbulent flow. Above the Reynolds number of 4,000, the flow is considered fully established turbulent flow.

For turbulent flow, the friction factor can be found mathematically through the use of the C. F. Colebrook equation or graphically from the chart developed by L. F. Moody (ASME 1944). This chart shows the relation between the friction factor, the Reynolds number, and the relative roughness of the pipe. The head loss is found by the main program by the following formula:

$$H = (1.653)\ (VPH^{1.748}\ \frac{(KV\ 100)}{0.001076}^{0.252}\ \frac{(2.307)}{D^{4.748}}$$

Where:

H = Head loss, ft
VPH = Flow rate, bbl/hr
KV = Kinematic viscosity, sq fps
D = Inside diameter (ID), in.

The variables of length, pipe diameter, and fluid velocity will impress the slope of the hydraulic gradient differentially depending on the value of the Reynolds number. Therefore, determination of the pressure requirements can be made only after the Reynolds number for each system in each variable combination is found.

With the hydraulic gradient in hand, the distance between pump sta-

Fig. 1

Fortran program for hydraulic calculations, profile drawings

```
                    WELCOME TO CROSS-COUNTRY PIPELINE.

       THIS PROGRAM WILL CALCULATE THE HEAD LOSS IN A CRUDE OIL PIPELINE
FOR BOTH LAMINAR AND TURBULENT FLOW. THE RESULTS WILL BE SIMILAR TO THOSE
TABULATED IN HYDRAULIC DATA HANDBOOKS . IT WILL ALSO LOCATE
PUMP STATIONS ALONG ANY ROUTE PROFILE BASED ON THE HIGH PRESSURE
LIMITATION YOU PLACE ON EACH PUMP STATION. IF YOU PREFER TO LOCATE
THE PUMP STATIONS, THIS PROGRAM WILL CALCULATE THE HEAD REQUIRED
TO DELIVER AT YOUR CHOSEN FLOW RATE. ALL CALCULATIONS INCLUDE FORTY
FEET OF HEAD BACK PRESSURE COMING INTO A STATION.
      WITH THIS PROGRAM YOU CAN VARY THE DIAMETER,VISCOSITY,GRAVITY,
STATION LOCATION, PRESSURE, FLOW RATE, LENGTH, OR TERRAIN AND SEE
HOW IT AFFECTS THE OTHER PARAMETERS, ALL AT THE TOUCH OF A BUTTON.
IN ADDITION TO THE NUMERICAL DATA OUTPUT, YOU MAY SUMMON THE
"TELLAGRAF" PROGRAM AND RUN IT USING THE FILE CALLED "HYDRAPRO"
AND IT WILL PRODUCE A HYDRAULIC PROFILE FOR YOU.

      JUST FOLLOW THE SIMPLE INSTRUCTIONS OUTLINED BELOW.
BUT BEFORE YOU START YOU MUST CHOOSE A ROUTE FROM THE TOPO MAPS
AND TABULATE THE ELEVATIONS IN FEET NO CLOSER THAN EVERY MILE. (SEE FOOTNOTE)

        FIRST ENTER THE NAME OF THIS PIPELINE USING " MARKS ON BOTH ENDS.
            NAME = "BLACK OIL PIPELINE"

        NEXT ENTER THE INTERNAL DIAMETER IN INCHES.
                    DIAMETER =    8.125

        ENTER THE PIPE LENGTH IN WHOLE MILES. (SEE FOOTNOTE)
                      LENGTH =   45.50
        ENTER THE GRAVITY IN DEGREES API.
                     GRAVITY =    14.5
        ENTER THE VISCOSITY AT THE OPERATING TEMPERATURE OF THE LINE.
                         SSU =   250.
        ENTER THE DISTANCE BETWEEN EACH ELEVATION POINT YOU'VE TAKEN.
                         GAP =      .5       IN WHOLE MILES.  (SEE FOOTNOTE)

        ENTER THE THRUPUTS YOU ARE INTERESTED IN, STARTING WITH THE
SMALLEST. USE UNITS OF BARRELS PER DAY.  IF YOU NEED LESS THAN TEN
LEAVE THE REST BLANK.
                      1  THRUPUT  =  30000.
                      2  THRUPUT  =       .
                      3  THRUPUT  =       .
                      4  THRUPUT  =       .
                      5  THRUPUT  =       .
                      6  THRUPUT  =       .
                      7  THRUPUT  =       .
                      8  THRUPUT  =       .
                      9  THRUPUT  =       .
                      10 THRUPUT  =       .

        IF YOU WANT THE PROGRAM TO LOCATE THE PUMP STATIONS FOR YOU,
FILL IN THE MAXIMUM PRESSURE ALLOWABLE AT EACH STATION IN PSI.
INCIDENTIALLY, YOU MAY VARY THIS MAXIMUM PRESSURE AT ANY PUMP STATION
AND SEE HOW IT AFFECTS THE LOCATION OF SUBSEQUENT STATIONS. IF THIS
IS THE FIRST TIME THROUGH YOU MAY NOT KNOW HOW MANY STATIONS WILL BE
NEEDED.  IN THIS CASE SET ALL THE PRESSURES AT SOME REASONABLE MAXIMUM.

                       STATION 1   =   700.   MAXIMUM PSI
                       STATION 2   =   700.   MAXIMUM PSI
                       STATION 3   =   700.   MAXIMUM PSI
                       STATION 4   =   700.   MAXIMUM PSI
                       STATION 5   =   700.   MAXIMUM PSI
                       STATION 6   =   700.   MAXIMUM PSI
                       STATION 7   =   700.   MAXIMUM PSI
                       STATION 8   =   700.   MAXIMUM PSI
                       STATION 9   =   700.   MAXIMUM PSI
                       STATION 10  =   700.   MAXIMUM PSI
                       STATION 11  =   700.   MAXIMUM PSI
                       STATION 12  =   700.   MAXIMUM PSI
                       STATION 13  =   700.   MAXIMUM PSI
                       STATION 14  =   700.   MAXIMUM PSI
                       STATION 15  =   700.   MAXIMUM PSI
                       STATION 16  =   700.   MAXIMUM PSI
                       STATION 17  =   700.   MAXIMUM PSI
                       STATION 18  =   700.   MAXIMUM PSI
                       STATION 19  =   700.   MAXIMUM PSI
                       STATION 20  =   700.   MAXIMUM PSI
```

```
        IF YOU KNOW THE PUMP STATION LOCATIONS AND WISH THE PROGRAM TO
CALCULATE THE REQUIRED HEAD FOR EACH OF YOUR THRUPUTS, PUT THE NUMBER
ONE HERE ( ). NOW FILL IN THE MILE POINTS OF EACH STATION BELOW. YOU
MUST ROUND OFF TO THE NEAREST WHOLE MILE. LEAVE ANY UNUSED STATIONS
BLANK.      (SEE FOOTNOTE)
                    MILE POINT      0.00  STATION 1 (FIRST ONE IS ALWAYS 0)
                    MILE POINT       .00  STATION 2
                    MILE POINT       .00  STATION 3
                    MILE POINT       .00  STATION 4
                    MILE POINT       .00 STATION 5
                    MILE POINT       .00 STATION 6
                    MILE POINT       .00 STATION 7
                    MILE POINT       .00 STATION 8
                    MILE POINT       .00 STATION 9
                    MILE POINT       .00 STATION 10
                    MILE POINT       .00 STATION 11
                    MILE POINT       .00 STATION 12
                    MILE POINT       .00 STATION 13
                    MILE POINT       .00 STATION 14
                    MILE POINT       .00 STATION 15
                    MILE POINT       .00 STATION 16
                    MILE POINT       .00 STATION 17
                    MILE POINT       .00 STATION 18
                    MILE POINT       .00 STATION 19
                    MILE POINT       .00 STATION 20

       FINALLY THE MOST TEDIOUS PART OF THIS PROGRAM. YOU MUST ENTER
THE LIST OF ELEVATIONS TAKEN FROM THE U.S.G.S. TOPOGRAPHIC MAPS.
THERE IS SPACE FOR 100 ENTRIES, YOU DON'T HAVE TO USE ALL OF THEM.
REMEMBER:1 THE ELEVATIONS MUST BE IN FEET.
         2 THERE MUST BE AN ELEVATION FOR THE ORIGIN AND THE TERMINAL.
         3 THIS PROGRAM WILL ASSUME THAT THE HORIZONTAL DISTANCE
           BETWEEN EACH POINT OF ELEVATION IS IDENTICAL  (AND EQUAL
           TO THE GAP WHICH YOU HAVE DEFINED ABOVE).
1  5750.0            26 6440.0            51  7690.0            76 6990.0
2  5770.0            27 6450.0            52  7680.0            77 6800.0
3  5760.0            28 6700.0            53  7580.0            78 6660.0
4  5760.0            29 6880.0            54  7590.0            79 6880.0
5  5760.0            30 6890.0            55  7500.0            80 6790.0
6  5720.0            31 6330.0            56  7470.0            81 6770.0

7  5700.0            32 6540.0            57  7470.0            82 6570.0
8  5680.0            33 6700.0            58  7470.0            83 6666.0
9  5660.0            34 6880.0            59  7400.0            84 6780.0
10 5670.0            35 6880.0            60  7380.0            85 6450.0
11 5640.0            36 6800.0            61  7400.0            86 6230.0
12 5640.0            37 6980.0            62  7380.0            87 5900.0
13 5640.0            38 7000.0            63  7400.0            88 5700.0
14 5680.0            39 7010.0            64  7490.0            89 5890.0
15 5990.0            40 7120.0            65  7450.0            90 6000.0
16 5790.0            41 7160.0            66  7400.0            91 6200.0
17 5880.0            42 7250.0            67  7330.0            92 6250.0
18 5830.0            43 7260.0            68  7320.0            93   00.0
19 6000.0            44 7260.0            69  7220.0            94   00.0
20 6110.0            45 7280.0            70  7150.0            95   00.0
21 6220.0            46 7300.0            71  7100.0            96   00.0
22 6220.0            47 7290.0            72  7100.0            97   00.0
23 6880.0            48 7500.0            73  7080.0            98   00.0
24 6540.0            49 7550.0            74  7030.0            99   00.0
25 6550.0            50 7690.0            75  6980.0           100   00.0
C
C    -----------------CREATED BY PURE TRANSPORTATION COMPANY
CC   -----------------WITH SPECIAL THANKS FOR THE TIMELY ASSISTANCE OF
C                     R.A.WATSON.
C
C FOOTNOTE:   YOU CAN USE FRACTIONS OF MILES FOR THE UNITS ON LENGTH,
C             ON THE GAP BETWEEN ELEVATIONS, OR ON THE STATION MILE
C             POINT LOCATIONS FOR GREATER PRECISION AS LONG AS THE
C             FOLLOWING REMAIN TRUE:
C
C             (LENGTH)/(GAP) = A WHOLE NUMBER
C             (STATION MILE POINT)/(GAP) = A WHOLE NUMBER
C
C        EXAMPLE:    LENGTH = 2.25 MILES
C                       GAP = 0.125 MILES
```

OGJ

tions along a pipeline route can be established if the maximum allowable pressure and the elevation profile of the route are known.

This is a graphic problem: Comparing point-by-point along the X axis, the Y values of the route profile, and the hydraulic gradient can reveal whether the allowable pressure is sufficient for each flow rate diameter, segment, and so forth.

Where this value is insufficient is the point to locate the next pump station. This process can be reiterated to the end of the line and must be repeated for each Reynolds number.

Conversely when the positions of the pump stations are known, the program can reverse the calculation and find the pressure requirements for each Reynolds number. Likewise the effect can be probed when one or more stations are eliminated, added, or altered in their maximum discharge pressure.

Graphically moving the hydraulic gradient over a height-distance plot of the pipeline route profile can reveal the distance traveled per foot of head at each point.

Program. This VS Fortran program will allow the engineer to vary any or all of these elements and see the results quickly. The program consists of a reading file, the main program, two writing files, and a job control file.

The engineer enters his parameters into the reading file. This file contains complete instructions on how to work the program (Fig. 1). The main program (Fig. 4) remains isolated from the user.

After reading the input data from the reading file, the main program will calculate the line fill for the system and print a table of the variables for the user's reference into one output file (Fig. 2) while filling another with graphing commands and notes which can be read by TELL-A-GRAF, a commercially available graphing software package.

The kinematic viscosity and dimensionless specific gravity are next found for the purpose of determining the Reynolds number. From this, the program decides the type of flow consistent with the input variables, prints its decision for the user (Fig. 2), and selects the subrouting to calculate the hydraulic gradient based on the option chosen.

The head loss due to friction is computed for each flow rate and stored in an array.

The main program will step off the route profile and calculate the total head required for each step, taking care never to let this requirement fall below zero or any previous demand

Fig. 2

Variables for hypothetical pipeline

```
                    "BLACK OIL PIPELINE"

 THIS OUTPUT IS FOR A PIPELINE HAVING THE FOLLOWING PARAMETERS:
  INSIDE DIAMETER =    8.125 INCHES
           LENGTH =    45.500 MILES
          GRAVITY =    14.50 DEGREES API
        VISCOSITY =    250.00 SSU
         LINEFILL =    15405.3 BARRELS

VELOCITY(FPS)    REYNOLDS NO.    FRICTION FACTOR   HEAD LOSS/MI   VOLUME(BPD)

    5.314          6139.672          0.0000           125.644        30000.

PUMP STA.#     HEAD(FT)     BHP       KILOWATTS    MILE PT.
    1          1420.79     357.95       297.        0.000
    2          1305.40     328.88       272.        9.000
    3          1316.68     331.73       275.       13.500
    4          1251.04     315.19       261.       20.000
    5          1150.30     289.81       240.       25.500
 THIS IS FOR A LINE RATE OF   30000.BPD
  AT THIS LINE RATE THE FLOW IS TURBULENT.
```

OGJ

Fig. 3

Program data for graph plot points

```
   GENERATE A PLOT.
 X AXIS LABEL TEXT IS "MILES".
 X PAGE IS 21.
 X AXIS LENGTH IS 16.
 Y PAGE IS 10.5.
 Y AXIS LENGTH IS 7.5.
 Y AXIS MIN IS 5076..
 Y AXIS LABEL TEXT IS "FEET".
 X GRID,Y GRID 2.
 FRAME.
 X TICKMARKS =     0.500.
 Y TICKMARKS = 5.
 TITLE  IS "BLACK OIL PIPELINE"
 MESSAGE 1 X = 10, Y = 103, UNITS = PLOT %.
 MESSAGE 1 BOX 0, CONNECT BL, HEIGHT .14, PAT 2, SWISS,
 TEXT IS "I(NSIDE DIAMETER) =  8.125 (IN)",
   "L(INEFILL) =     15.4 M(BBLS)".
 LEGEND X ORIGIN = 105, LEGEND Y ORIGIN = 0.
 LEGEND UNITS = PLOT %.
 TITLE STYLE IS SWISS.
 INPUT DATA.
   "TERRAIN"
           0   5750.
        0.50   5770.      1.00   5760.      1.50   5760.
        2.00   5760.      2.50   5720.      3.00   5700.
        3.50   5680.      4.00   5660.      4.50   5670.
        5.00   5640.      5.50   5640.      6.00   5640.
        6.50   5680.      7.00   5990.      7.50   5790.
        8.00   5880.      8.50   5830.      9.00   6000.
        9.50   6110.     10.00   6220.     10.50   6220.
       11.00   6880.     11.50   6540.     12.00   6550.
       12.50   6440.     13.00   6450.     13.50   6700.
       14.00   6880.     14.50   6890.     15.00   6330.
       15.50   6540.     16.00   6700.     16.50   6880.
       17.00   6880.     17.50   6800.     18.00   6980.
       18.50   7000.     19.00   7010.     19.50   7120.
       20.00   7160.     20.50   7250.     21.00   7260.
       21.50   7260.     22.00   7280.     22.50   7300.
       23.00   7290.     23.50   7500.     24.00   7550.
       24.50   7690.     25.00   7690.     25.50   7680.
       26.00   7580.     26.50   7590.     27.00   7500.
       27.50   7470.     28.00   7470.     28.50   7470.
       29.00   7400.     29.50   7380.     30.00   7400.
       30.50   7380.     31.00   7400.     31.50   7490.
       32.00   7450.     32.50   7400.     33.00   7330.
       33.50   7320.     34.00   7220.     34.50   7150.
       35.00   7100.     35.50   7100.     36.00   7080.
       36.50   7030.     37.00   6980.     37.50   6990.
       38.00   6800.     38.50   6660.     39.00   6880.
       39.50   6790.     40.00   6770.     40.50   6570.
       41.00   6666.     41.50   6780.     42.00   6450.
       42.50   6230.     43.00   5900.     43.50   5700.
       44.00   5890.     44.50   6000.     45.00   6200.
       45.50   6250.     46.00      0.     46.50      0.
   "  30000. BPD"
            0.00          7170.79
            9.00          6040.00
            9.00          7305.39
           13.50          6740.00
           13.50          8016.68

           20.00          7200.00
           20.00          8411.04
           25.50          7720.00
           25.50          8830.30
           45.50          6317.42
           45.50          6250.00
   END OF DATA.
 CURVE 1 SHADE PAIR IS-1,               SHADE COLOR IS GREEN.
 CURVE 1 TEXTURE IS 13.
 GO.
 SEND.
 **FILE**
```

OGJ

nor rise above the stipulated maximum allowable pressure. The total head required includes friction loss, elevation, and net positive suction head.

A pump station is located where this calculation can go no further, and the calculation is repeated to the next pump station. The head required at each is stored in the computer memory. Using this value and other variables, the program calculates the brake horsepower and Kilovolt-ampere for each station (Fig. 2).

The numbers representing pressure and elevation-vs.-distance for each flow rate are stored in two column arrays and the values of elevation alone vs. distance are stored in a similar array. These arrays are then printed into a file which is used as the X and Y data input for the TELL-A-GRAF software package (Fig. 3).

The Union Oil Co. uses IBM 3278 or 3279 terminals to create this program, but viewing the graphics requires an IBM program symbols card. The mainframe is an IBM 3033 or 3081, although an IBM 370 would be sufficient. The disc pac is an IBM 3380. Printing is done on a IBM 3800 laser printer for alphanumeric outputs and a Xerox Versatec plotter for the

Fig. 4

Main program for cross-country pipeline

```
C        CROSS-COUNTRY PETROLEUM PIPELINE IS A PROGRAM WRITTEN IN
C        FORTRAN WHICH PRODUCES HYDRAULIC DATA AND PROFILES.
C
      REAL NAME(8)
      DIMENSION R(50),F(50),H(50),V(50),MP(20),HEAD(100),DIS(100),
     +ELAV(100),HED(100),MPI(20),XY(40,2),Y(20),DEAD(100),
     +LOK(20),HA(100),HH(100),PAP(100)
      INTEGER DOT
      REAL LEN,KV,MP,MPI,LOC
      DIMENSION E(100),LOC(20),PSI(20),VOL(10),FLO(10)
C        FROM HERE TO LINE 56 ARE THE STATEMENTS WHICH READ THE INPUT
C     VARIABLES FROM THE 'PROFILE' FILE.
      DO 40 I=1,22
      READ(8,50) DUMMY
   40 CONTINUE
   50 FORMAT(A1)
C     REAL  NAME(8)
      READ (8,58)(NAME(I),I=1,8)
   58 FORMAT(19X,8(A4))
      READ(5,50) DUMMY
      READ(8,50) DUMMY
      READ(8,51) DIA
   51 FORMAT(31X,F8.3)
      READ(8,50) DUMMY
      READ(8,50) DUMMY
      READ(8,52) LEN
   52 FORMAT(31X,F9.3)
      READ(8,50) DUMMY
      READ(8,53) GRAV
   53 FORMAT(31X,F9.3)
      READ(8,50) DUMMY
      READ(8,53) SSU
      READ(8,50) DUMMY
      READ(8,53) GAP
      READ(8,50) DUMMY
      READ(8,50) DUMMY
      READ(8,50) DUMMY
      READ(8,50) DUMMY
      DO 41 I=1,10
      READ(8,54) VOL(I)
   41 CONTINUE
   54 FORMAT(31X,F9.2)
      DO 42 I=1,8
      READ (8,50) DUMMY
   42 CONTINUE
      DO 43 I=1,20
      READ(8,53) PSI(I)
   43 CONTINUE
      READ(8,50) DUMMY
      READ(8,50) DUMMY
      READ(8,50) DUMMY
      READ(8,50) DUMMY
      READ(8,55) DOT
   55 FORMAT(10X,I1)
      READ(8,50) DUMMY
      READ(8,50) DUMMY
      DO 44 I=1,20
      READ(8,53) LOC(I)
   44 CONTINUE
      DO 45 I=1,10



      READ(8,50) DUMMY
   45 CONTINUE
      DO 46 I=1,25
      READ(8,56) E(I),E(I+25),E(I+50),E(I+75)
   46 CONTINUE
   56 FORMAT(3X,F5.0,14X,F5.0,14X,F5.0,15X,F5.0)
C
C        HERE WE CALCULATE THE LINEFILL.
      FILL = 3.14159 * ((DIA /24)**2) * LEN * 5280 *7.48/42
      PILL = FILL/1000
C
C        HERE WE PRINT THE INPUT VARIABLES FOR THE VIEWERS REFERENCE
      WRITE(9,500) (NAME(I),I=1,8),DIA,LEN,GRAV,SSU,FILL
      WRITE(6,500) (NAME(I),I=1,8),DIA,LEN,GRAV,SSU,FILL
  500 FORMAT(15X,8(A4),//,2X,'THIS OUTPUT IS FOR A PIPELINE HAVING '
     +'THE FOLLOWING PARAMETERS:',
     +/,  '    INSIDE DIAMETER = ',F8.3,' INCHES',
     +/,  '             LENGTH = ',F9.3,' MILES',
     +/,  '            GRAVITY = ',F8.2,' DEGREES API',
     +/,  '          VISCOSITY = ',F8.2,' SSU',
     +/,  '           LINEFILL = ',F9.1,' BARRELS',//)
C
      SMALL = E(1)
      DO 562 I =1,100
      IF(E(I).EQ.0) GO TO 562
      IF(SMALL.GT.E(I)) SMALL = E(I)
  562 CONTINUE
      GALL = SMALL*.9
C      HERE WE PRINT THE GRAPHING COMMANDS FOR THE 'TELAGRAF PROGRAM'
      WRITE(10,300)  GALL,     GAP,(NAME(IP), IP=1,8),DIA,PILL
  300 FORMAT(2X ,'GENERATE A PLOT.',/,
     +'X AXIS LABEL TEXT IS "MILES".',/,
     +'X PAGE IS 21.',/,
     +'X AXIS LENGTH IS 16.',/,
     +'Y PAGE IS 10.5.',/,
     +'Y AXIS LENGTH IS 7.5.',/,
     +'Y AXIS MIN IS ',F5.0,'.',/,
     +'Y AXIS LABEL TEXT IS "FEET".',/,
     +'X GRID,Y GRID 2.',/,
     +'FRAME.',/,
     +'X TICKMARKS = ',F9.3,'.',/,
     +'Y TICKMARKS = 5.',/,
     +'TITLE  IS ',8(A4),'.',/,
     +'MESSAGE 1 X = 10, Y = 103, UNITS = PLOT %.',/,
     +'MESSAGE 1 BOX 0, CONNECT BL, HEIGHT .14, PAT 2, SWISS.',/,
     +'TEXT IS "I(NSIDE DIAMETER) = ',F6.3,' (IN)",',/,
     +'  "L(INEFILL) = ',F8.1,' M(BBLS)".',/,
     +'LEGEND X ORIGIN = 105, LEGEND Y ORIGIN = 0.',/,
     +'LEGEND UNITS = PLOT %.',/,
     +'TITLE STYLE IS SWISS.',/,'INPUT DATA.')
C
C        HERE WE CALCULATE THE KINEMATIC VISCOSITY IN SQ. FT. PER SEC.,
C          THE DIMENSIONLESS SPECIFIC GRAVITY, AND THE DIAMETER IN FEET.
      IF(SSU.LE.100) KV = (.00226*SSU - 1.95/SSU) *.001076
      IF(SSU.GT.100) KV = (.0022*SSU - 1.35/SSU)* .001076
      SG = 141.5 * (1 /(131.5 + GRAV))
      D = DIA/12
C
C     DIMENSION R(50),F(50),H(50),V(50)
      WRITE(9,209)
```

```
      WRITE(6,209)
  209 FORMAT(1X,'VELOCITY(FPS)',5X,'REYNOLDS NO.',5X,'FRICTION FACTOR',
     +5X,'HEAD LOSS/MI',5X,'VOLUME(BPD)',/)
C
C        HERE WE CALCULATE THE REYNOLDS NUMBER AND DETERMINE THE TYPE
C           OF FLOW.
      DO 6 I=1,10
      IF(VOL(I).EQ.0)  GO TO 6
      V(I) = VOL(I) * .0000012 /(D **2.)
      R(I) = V(I) * D / KV
      IF(R(I).LE.2200) FLO(I) = 1
      IF(R(I).GE.4000) FLO(I) = 2
      IF((R(I).GT.2200).AND.(R(I).LT.4000)) FLO(I) = 3
      IF(R(I).GT.2200) GO TO 100
C
C           HERE WE CALCULATE THE HEAD LOSS FOR LAMINAR FLOW.
      F(I) = 64/R(I)
      H(I) = F(I) * (V(I) **2.) * 82.0538 / D
C
      WRITE(6,210)V(I),R(I),F(I),H(I),VOL(I)
      WRITE(9,211)V(I),R(I),F(I),H(I),VOL(I)
  210 FORMAT(3X,F8.3,5X,F16.3,6X,F8.4,10X,F9.3,6X,F8.0,/)
  211 FORMAT(3X,F8.3,5X,F16.3,6X,F8.4,10X,F9.3,6X,F8.0,/)
  520 FORMAT(3X,'AT THIS LINE RATE THE FLOW IS LAMINAR.',//)
  521 FORMAT(3X,'AT THIS LINE RATE THE FLOW IS TURBULENT.',//)
  522 FORMAT(3X,'AT THIS LINE RATE THE FLOW IS IN THE TRANSITION ZONE,'
     +,/,' SO THE PRESSURES INDICATED ABOVE ARE APPROXIMATE.',//)
C
      IF(R(I).LE.2200) GO TO 6
C
C           HERE WE CALCULATE THE HEADLOSS FOR TURBULENT FLOW.
  100 VPN = VOL(I) / 24
      H(I)=(1.65*(VPN **1.748))*((KV*100)/.001076)**.252)*2.307
     +/( DIA **4.748)* .97
      F(I) =     0.0
C
      WRITE(6,210)V(I),R(I),F(I),H(I),VOL(I)
      WRITE(9,211)V(I),R(I),F(I),H(I),VOL(I)
C
    6 CONTINUE
C
C        HERE WE WRITE THE TERRAIN CURVE INTO X AND Y COORDINATES.
      WRITE(10,240) E(1)
  240 FORMAT(2X,'"','TERRAIN','"',/, 10X,'0',5X, F8.0)
      DO 19 I=2,100
      PAP(I) = GAP * FLOAT(I-1)
      IF(I.EQ.1) PAP(I)=GAP
   19 CONTINUE
      DO 108 I=2,98,3
      IF(E(I).EQ.0) GO TO 108
      WRITE(10,244) PAP(I),E(I),PAP(I+1),E(I+1),PAP(I+2),E(I+2)

  108 CONTINUE
C
  244 FORMAT(5X,F8.2,5X,F5.0,5X,F8.2,5X,F5.0,5X,F8.2,5X,F5.0)
C
C     DIMENSION MP(20),HEAD(100),DIS(100),ELAV(100),HED(20),MPI(20),
C    +XY(40,2),Y(20), DEAD(100),LOK(20), HA(100), HH(100)
C
C           HERE WE START CALCULATING THE HEAD REQUIRED DUE TO FRICTION



C           AND ELEVATION.
C           J---INDEXES THE FLOW RATES
C           L & LL---INDEX THE DISTANCES, ASSOCIATED ELEVATIONS AND
C           THE HEAD REQUIRED FOR EACH STEP.
C           K---- INDEXES THE PUMP STATIONS.
      DO 7 J= 1,10
      DO 13 I =1,100
      ELAV(I) = E(I)
C     ELAV(I) = 100
   13 CONTINUE
C
      IF(VOL(J).EQ.0)  GO TO 7
      WRITE(9,330)
      WRITE(6,330)
C
C
C           BETWEEN HERE AND LINE 103 WE CALCULATE THE HEAD IF THE
C           STATION LOCATIONS ARE THE LIMITING FACTOR.
      DO 8 K=1,20
      IF(DOT.EQ.0) GO TO 103
      IF (LOC(K+1).EQ.0) LOC(K+1) = LEN
      IMP = (LOC(K+1)- LOC(K))/GAP
      DO 30 LL=1, IMP
      EHEAD = ELAV(LL+1)-ELAV(1)
      DEAD(LL) = ((GAP*FLOAT(LL))*H(J))+EHEAD+40
      IF(DEAD(LL).LT.0) DEAD(LL)=0
      IF(IMP.EQ.1) GO TO 80
      IF (LL.EQ.1) GO TO 30
      IF(DEAD(LL).LT.DEAD(LL-1)) DEAD(LL)=DEAD(LL-1)
   30 CONTINUE
      IF(DOT.EQ.1) GO TO 80
C
C
C
C
C
C
C
C
C
C           BETWEEN HERE AND LINE 80 WE CALCULATE THE HEAD IF THE
C           STATION PRESSURES ARE THE LIMITING FACTOR.
  103 DO 9 L =1,100
      DIS(L) =   FLOAT(L)    * GAP
      IF(K.EQ.1) MP(K) = DIS(L)
      IF(K.EQ.1) GO TO 102
      MP(K) = DIS(L) + MP(K-1)
  102 CONTINUE
C
      EHEAD = ELAV(L+1) -ELAV(1)
      HIP =(PSI(K)* 2.03 /SG)- (DIS(1)*H(J) +40)
      HEAD(L) =DIS(L) * H(J) + EHEAD + 40
      IF(HEAD(L).LT.0) HEAD(L) = 0
      IF(L.EQ.1) GO TO 18
      IF(HEAD(L).LT.HEAD(L-1)) HEAD(L) = HEAD(L-1)
   18 IF(MP(K).GE.LEN)  GO TO 10
      PIP = HIP * .1
      ZIP = HIP - HEAD(L)
      IF(ZIP  .LT.  PIP ) GO TO 10
    9 CONTINUE
   10 CONTINUE
C
```

```
C        HERE WE EXCHANGE VARIABLE NAMES FOR AN EASY WRITE FORMAT.
      IF(K.GT.1) MPI(K)= MP(K-1)
      HED(K) = HEAD(L)
      IF(K.EQ.1) MPI(1) = 0.00
   80 IF(DOT.EQ.0) GO TO 104
      IF(DOT.EQ.1) MPI(K) = LOC(K)
      IF(DOT.EQ.1) HED(K) = DEAD(IMP)
C
  104 CONTINUE
C
C        HERE WE CALCULATE BRAKE HORSEPOWER AND KILOWATTS.
C
      BHP =       VOL(J)/24  *  42/60 *  HED(K)*SG  / 3960/.85
C
C
      WATTS = BHP*.7457 / .90
C
C     IF(DOT.EQ.1) WRITE(6,212) KAP,HED(KAP),BHP,WATTS,MPI(KAP)
C     IF(DOT.EQ.1) WRITE(9,212) KAP,HED(KAP),BHP,WATTS,MPI(KAP)
C     IF(DOT.EQ.1) GO TO 910
      WRITE(6,212) K,HED(K),BHP,WATTS,MPI(K)
      WRITE(9,212) K,HED(K),BHP,WATTS,MPI(K)
  910 CONTINUE
  330 FORMAT(1X,'PUMP STA.#',5X,'HEAD(FT)',5X,'BHP',8X,'KILOWATTS',4X,
     +'MILE PT.')
  212 FORMAT(5X,I2,6X,F10.2,4X,F8.2,4X,F6.0,4X,F9.3)
C
C
C
C
      IF(MP(K).GE.LEN)  GO TO 11
C     IF((DOT.EQ.1).AND.(LOK(3).EQ.LEN))GO TO 11
C
C          HERE WE SHIFT THE ARRAY OF ELEVATIONS FOR THE NEXT PUMP
C             STATION CALCULATIONS.
  906 CONTINUE
      DO 12  I=1,100
C
      ELAV(I) = ELAV(I+L)
C
C
   12 CONTINUE
C
      IF(LOC(K+1).EQ.LEN) GO TO 11
    8 CONTINUE
C
   11 WRITE(9,215) VOL(J)
      WRITE(6,215) VOL(J)
      IF(FLO(J).EQ.1) WRITE(6,520)
      IF(FLO(J).EQ.1) WRITE(9,520)
      IF(FLO(J).EQ.2) WRITE(9,521)
      IF(FLO(J).EQ.2) WRITE(6,521)
      IF(FLO(J).EQ.3) WRITE(9,522)
      IF(FLO(J).EQ.3) WRITE(6,522)
C
  215 FORMAT(2X,'THIS IS FOR A LINE RATE OF',F8.0,'BPD')
C
C          THE REST OF THE PROGRAM MASSAGES THE HEADS, ELEVATIONS,
C             AND MILEPOINTS INTO X & Y COORDINATES FOR THE TELAGRAF
C             PROGRAM.



      DO 20 M=2,20
      MM   =    MPI(M)/GAP
      IF(MPI(M).EQ.0)  GO TO 29
      Y(M) = E(MM+1)
      GO TO 20
   29 KK= LEN/GAP
      Y(M)= E(KK+1)
      MPI(M) = LEN
   20 CONTINUE
C
      XY(1,1) = 0.0
      XY(1,2) = HED(1) + E(1)
C
      DO 22 II = 2,20
      IK = 2*II - 1
      XY(IK,1) = MPI(II)
      XY(IK,2) = HED(II) + Y(II)
      IF(MPI(II).EQ.LEN) GO TO 61
   22 CONTINUE
C
   61 DO 21 JJ = 2,40,2
      KJ = JJ -(JJ-2)/2
      XY(JJ,1) = MPI(KJ)
      XY(JJ,2) = XY(JJ-1,2) - ((MPI(KJ) - MPI(KJ-1))*H(J))
      IF(MPI(KJ).EQ.LEN) GO TO 60
   21 CONTINUE
C
   60 WRITE(10,224) VOL(J)
      DO 23 KM = 1,40
      IF((XY(KM,1).EQ.0).AND.(XY(KM,2).EQ.0))  GO TO 23
      WRITE(10,223) XY(KM,1), XY(KM,2)
   23 CONTINUE
C
  223 FORMAT(5X,F10.2,5X,F10.2)
  224 FORMAT(2X,'"',F8.0,' BPD','"')
C
      DO 24 MK=2,20
      HEAD(MK)=0
      Y(MK)= 0
      MP(MK)  = 0
      HED(MK) = 0
      MPI(MK) = 0
   24 CONTINUE
C
      DO 25  LIK=1,40
      XY(LIK,1)=0
      XY(LIK,2)=0
   25 CONTINUE
      KAP = 0
    7 CONTINUE
C
      WRITE(10,218)
  218 FORMAT(2X,'END OF DATA.',/,'CURVE 1 SHADE PAIR IS-1,
     + SHADE COLOR IS GREEN.',/,
     +'CURVE 1 TEXTURE IS 13.',/,'GO.',/,'SEND.',/,'**FILE**')
C
  101 CONTINUE
      GO TO 904
  901 WRITE(9,902) K
      WRITE(6,902) K
  902 FORMAT(3X,'ALLOWABLE PRESSURE AT STATION ',I4,' IS TOO LOW TO',
     +'REACH THE NEXT STATION.  GO BACK TO THE INPUT FILE AND BE ',/,
     +'MORE GENEROUS.')
  904 CONTINUE
      END
```

OGJ

Fig. 5

Profile for Black Oil Pipeline



graphical outputs.

Hypothetical pipeline. The example shown in the figures is for a fictitious project entitled "Black Oil Pipeline." The pipe diameter is a nominal 8 in., the length is 45½ miles, the gravity of the crude oil to be moved on this line is 14.5° API, and the viscosity at the operating temperature (@ 65° F.) is 250 SUS.

- This title and four parameters constitute the operator's first five data entries to Fig. 1. It is important to keep the decimal point at the same location on the screen through all of the data input because the main program will search a certain field for this data and some information could be omitted.
- The sixth entry is the horizontal distance between each elevation data point along the pipeline route called the "gap." In this example, an elevation reading was taken every half mile. This will be explained further below.
- The next group of inputs defines the pumping rates which the main program will use to calculate friction loss. On the Black Oil Pipeline, 30,000 b/d is the only flow rate of interest. If the other spaces were used, the program would calculate each in turn and plot them all on the same graph (Fig. 5).
- The next two columnar fields (Fig. 1) of input data are used separately.

The first sets the maximum operating pressure at the pump stations. This pressure must be set if the program is used to establish the location of pump stations based on the hydraulics of the line.

Since the Black Oil Pipeline is a project in the planning stages, a maximum pressure has been set at 700 psi for all possible pump stations (20). Later we could change the pressure limit at some of the stations to reduce the power required.

For example (not shown), if the pressure at pump station No. 4 were increased by 150 psi, pump station No. 5 could be located at mile point 31 and require only about 300 psi to move the crude oil to the end of the line.

The second column of input data stipulates the mile point locations of each pump station. If the number "1" is placed in the parentheses and mile point information is entered in the column, the main program will ignore the maximum pressures given previously and perform the hydraulics based on station locations only.

This mode is useful in analyzing existing pipelines or planned pipelines in which pump station location is determined by factors other than hydraulics, such as geography, or proximity to roads or power sources. In the Black Oil Pipeline, the station locations are undetermined. This method is not used, therefore, and the parentheses are left blank.

- The final data entries are the elevations along the pipeline route. This information can be obtained from the ground surveys, but in the planning stages, the route is more often selected by studying aerial photographic, Bureau of Land Management maps, and U.S. Geological Survey topographic maps.

Whatever the source of information, the elevation in feet along the route must be tabulated. Each elevation must be taken at the same horizontal distance from the previous one.

If working from a topographical map, set the dividers at a certain gap and tabulate an elevation at each step of the divider. The first elevation will be at the pipeline origin and the last will be the pipeline terminal.

The note of Fig. 1 reveals that fractional miles can be used for the pipeline length, the gap, and the pump station mile point locations. The program uses the quotient of length/gap and station mile point/gap as the indices of several do-loops; therefore, quotient must always be a whole number.

In the Black Oil Pipeline example, this is illustrated by a length of 45.5 miles and a gap between elevations of one-half mile, thus $45.5/0.5 = 91$.

Incidentally, the length used is the horizontal length of the route. The program calculates the actual pipe length and uses this value in determining head loss. ▪

The author . . .


Dickman

Thomas A. Dickman is a pipeline engineer for Pure Transportation Co., the pipeline subsidiary for Union Oil Co. of California. Recent projects include Union's shale oil pipelines in Colorado and heavy oil pipelines in Wyoming. He received a BS in mechanical engineering from the University of Illinois. He is a member of the American Society of Mechanical Engineers.

BOOKS

Hydraulic Handbook, *by R.H. Warring. Published by Gulf Publishing Co.—Book Division, Box 2608, Houston, Tex., 77001. 467 pp., $72.50.*

This is the eighth edition of this reference book. It is fully updated, with current information on interface technology, miniaturization, computerization, super durable materials, automatic monitoring, and controls.

Microbial Enhanced Oil Recovery, *edited by J.E. Zajic, D.G. Cooper, T.R. Jack, and N. Kosaric. Published by PennWell Publishing co.—Book Division, P.O. Box 21288, Tulsa, Okla., 74121. 183 pp., $37.50.*

This collection of papers presented at the Canadian Microbial Enhanced Oil Recovery Symposium displays the latest state-of-the-art techniques by top international experts in this field.

The editors were all at one time in the biochemistry engineering group of the University of Western Ontario, London, Ont. Their investigations and research into microbial leaching and microbial geoseparation processes resulted in patents and over 60 publications.